BUIDLING HEATING SYSTEM DESIGN

A consulting engineering firm is designing the heating and cooling systems for a high-rise office building in downtown Seattle. Steam is to be used as a heat source for these systems and is supplied to the mechanical room of the building via a vertical pipe. The mechanical room is on the very top story of the building. The steam enters the pipe at the bottom as a saturated vapor at 30 psia. Each story of the building is 12 ft high. In the mechanical room, the steam pipe must run to the top of the story. In any single story, the heat loss from the vertical steam pipe is anticipated to be 15 Btu/lbm and the steam pressure drop is 0.5 psia.

a. Determine the quality of the steam at the top of the 5^{th} story of the building.

b. Using EES, develop a plot that shows the number of stories in the building (vertical axis) vs. the quality of the steam on each story (horizontal axis).

c. A mechanical engineer has determined that the steam is no longer useful for heating or cooling purposes once its quality drops below 40%. How many stories tall can the building be?



Given: A high-rise building.



Each story has a height of 12 ft.
Each story has a ΔP of 0.5 psia
Each story has a heat loss of 15 $\frac{Btu}{lbm}$

$X_{min} = \underline{40\%}$

Find: (a) The quality of the steam on the 5^{th} story
(b) n vs. x_n
(c) How tall can the building be?

Solution: Consider the pipe at any story,

$P_2 = P_1 - n\Delta P$ ← 0.5 psi



The First Law applied to this system is, (assume $\Delta KE \cong 0$)

$$\dot{Q} - \cancel{\dot{W}}^{\,0} + \dot{m}_1\left(h_1 + \cancel{\frac{V_1^2}{2g_c}} + \cancel{\frac{g}{g_c}z_1}\right) - \dot{m}_2\left(h_2 + \cancel{\frac{V_2^2}{2g}} + \frac{g}{g_c}z_2\right) = \cancel{\frac{dE_{cv}}{dt}}^{\,0} \quad \text{SSSF}$$

($z_1 = 0$)

Then,
$$\dot{Q} + \dot{m}_1 h_1 - \dot{m}_2\left(h_2 + \frac{g}{g_c}z_2\right) = 0$$

Continuity: $\dot{m}_1 - \dot{m}_2 = 0 \rightarrow \dot{m}_1 = \dot{m}_2 \equiv \dot{m}$

Therefore
$$\frac{\dot{Q}}{\dot{m}} + h_1 - \left(h_2 + \frac{g}{g_c}z_2\right) = 0$$

This equation can be written as,

$$q + h_1 - \left[h_2 + \frac{g}{g_c}(n\Delta z)\right] = 0 \qquad \Delta z = 12\ \text{ft}$$

If the number of building stories is known, the only unknown above is h_2.
The second property required to fix the state is pressure.

$$P_2 = P_1 - n\Delta P \qquad \text{where } \Delta P = 0.5\ \text{psia}$$

Now, P_2, h_2 are known! Therefore x_2 can be found.

EES

```
"GIVEN:  A high-rise building being heated using steam"

    f$ = 'steam_iapws'
    P_1 = 30[psia]          "Pressure at the bottom of the steam pipe"
    x_1 = 1                 "quality at the bottom of the steam pipe"
    z_1 = 0[ft]             "entrance of the pipe in the building"
    q = -15[Btu/lbm]        "heat loss per story"
    DELTAz = 12[ft]         "height of each story in the building"
    DELTAP = 0.5[psi]       "pressure drop of the steam per story"
    g = 32.174[ft/s^2]
    g_c = 32.174[lbm-ft/lbf-s^2]

"FIND: (a) the quality of the steam on the 5th floor
       (b) a plot of building height vs. steam quality
       (c) how many stories tall can the building be?"

"SOLUTION:"
$IFNOT Parametric Study
    n = 5   "number of stories"
$ENDIF

"The First Law applied to the pipe from the bottom entrance (1) to the point
 in question (2) is given by,"
    q_12 + (h_1 + pe_1) - (h_2 + pe_2) = 0
    h_1 = enthalpy(f$,P=P_1,x=x_1)
    q_12 = n*q

"The potential energies at states 1 and 2 are given by,"
    pe_1 = (g/g_c)*z_1*convert(ft-lbf/lbm,Btu/lbm)
    pe_2 = (g/g_c)*n*DELTAz*convert(ft-lbf/lbm,Btu/lbm)

"This allows the enthalpy, h_2, to be found.  The pressure can also be found
 at state 2.  This fixes the state.  Therefore, the quality can be found!"
    P_2 = P_1 - n*DELTAP
    x_2 = quality(f$,P=P_2,h=h_2)
```

GIVEN: A high-rise building being heated using steam

$f\$$ = 'steam$_{iapws}$'

$P_1 = 30$ [psia] *Pressure at the bottom of the steam pipe*

$x_1 = 1$ *quality at the bottom of the steam pipe*

$z_1 = 0$ [ft] *entrance of the pipe in the building*

$q = -15$ [Btu/lbm] *heat loss per story*

Δz = 12 [ft] *height of each story in the building*

ΔP = 0.5 [psi] *pressure drop of the steam per story*

g = 32.174 [ft/s^2]

g_c = 32.174 [lbm-ft/lbf-s^2]

FIND: (a) the quality of the steam on the 5th floor
(b) a plot of building height vs. steam quality
(c) how many stories tall can the building be?

SOLUTION:

$n = 5$ *number of stories*

The First Law applied to the pipe from the bottom entrance (1) to the point in question (2) is given by,

$$q_{12} + h_1 + pe_1 - [h_2 + pe_2] = 0$$

$$h_1 = \mathbf{h}[f\$, P = P_1, x = x_1]$$

$$q_{12} = n \cdot q$$

The potential energies at states 1 and 2 are given by,

$$pe_1 = \frac{g}{g_c} \cdot z_1 \cdot \left| 0.001285067 \cdot \frac{\text{Btu/lbm}}{\text{ft–lbf/lbm}} \right|$$

$$pe_2 = \frac{g}{g_c} \cdot n \cdot \Delta z \cdot \left| 0.001285067 \cdot \frac{\text{Btu/lbm}}{\text{ft–lbf/lbm}} \right|$$

This allows the enthalpy, h_2, to be found. The pressure can also be found at state 2. This fixes the state. Therefore, the quality can be found!

$$P_2 = P_1 - n \cdot \Delta P$$

$$x_2 = \mathbf{x}[f\$, P = P_2, h = h_2]$$

SOLUTION

Unit Settings: Eng F psia mass deg

ΔP = 0.5 [psi]
f$ = 'steam_iapws'
g_c = 32.17 [lbm-ft/lbf-s^2]
h_2 = 1089 [Btu/lbm]
pe_1 = 0 [Btu/lbm]
P_1 = 30 [psia]
q = -15 [Btu/lbm]
x_1 = 1
z_1 = 0 [ft]
Δz = 12 [ft]
g = 32.17 [ft/s^2]
h_1 = 1164 [Btu/lbm]
n = 5
pe_2 = 0.0771 [Btu/lbm]
P_2 = 27.5 [psia]
q_{12} = -75 [Btu/lbm]
x_2 = 0.9227

No unit problems were detected.

KEY VARIABLES
x_2 = 0.9227 *Quality of the steam on the 5th floor*



REFLECTION:

EES really makes this problem EESy! The quality can be found with a simple property call given P and h! The alternative is a lengthy hand-calculation $x = (h - h_f)/h_{fg}$. The fact that EES does this quickly, and is capable of filling in a parametric tables makes short order of the calculations!

Thermodynamics is everywhere! Here, we see that thermodynamics tells us how tall the building can be based on the requirement of steam with a minimum quailty of 40%. Very cool!

Based on what we see in the plot, the building can be no taller than **40 stories**, otherwise the quality drops below the 40% limit.